
/* Program: Chiptemperatur */ 
 
/* Hvilken temperatur har du på chippen? 
  
 Vi bruger en intern temperatur føler i chippen 
  
 10. marts 2013 
 Erik Christiansen alias OZ7EC 
 */ 
 
 
/* Display forbindelser 
 pin 11 LCD pin D4 til Arduino digital pin 5 
 pin 12 LCD pin D5 til Arduino digital pin 4 
 pin 13 LCD pin D6 til Arduino digital pin 3 
 pin 14 LCD pin D7 til Arduino digital pin 2 
 pin 3 LCD til stel = fuld kontrast til vore gamle øjne 
 pin 4  LCD pin RS til Arduino digital pin 12 
 pin 6 LCD Enable pin til Arduino digital pin 11 
 pin 5 LCD R/W pin til stel    */ 
 
/* Backlight kræver 150 mA ved 5 V, som stabilisatoren på 
 Arduino Uno R3 IKKE kan bære. Den er en SMD kreds. 
 Derfor separat 5 volt stabilisator 7805 med køleplade 
 fødet direkte af V(in) på Arduino 
 V(in) er den rå spænding fra net tilslutningen, IKKE fra USB 
 Som Vin bruges en af de sorte net forsyninger, 
 plus i midten 
 en 6 Vdc er nok., en 12 volt giver varm regulator 
 */ 
 
/*----------------------------------------------------- 
 * Her starter definitionerne 
 -------------------------------------------------------*/ 
 
/* Vi referer til biblioteker */ 
#include <LiquidCrystal.h> 
 
#if defined(ARDUINO) && ARDUINO >= 100 
#include "Arduino.h" 
#else 
#include "WProgram.h" 
#endif 
 
//definer variable og konstanter 
#define karaktererperlinie  16 
#define  antallinier  2 
 
 
/********************************************************* 



 * Fortæl bibliotek hvilke Arduino pinde hører til hvilke 
 * funktioner i displayet 
 ********************************************/ 
LiquidCrystal lcd(12,11,5,4,3,2);   // Arduino's 
 
double temperatur; 
/*---------------------------------------------- 
 * slut på definitioner 
 ------------------------------------------------*/ 
 
 
/*-------------------------------------------- 
 *  Start på SETUP 
 ----------------------------------------------*/ 
 
 
/******************************************** 
 * Skriv en velkomst på display 
 ********************************************/ 
void setup() 
{ 
  lcd.begin (karaktererperlinie,antallinier); 
  lcd.setCursor(0,0);  // øverste linie, første position 
  lcd.print ("Chip Temperatur ");  // skriv på øverste linie 
  lcd.setCursor(0,1);  // første position, 2. linie 
  lcd.print ("       Grader C "); // skriv på nederste linie 
} 
/*-------------------------------------------- 
 *  Slut på SETUP 
 *--------------------------------------------*/ 
 
 
 
 
/*-------------------------------------------- 
 *  Start på programkontrol 
 ----------------------------------------------*/ 
void loop() 
{ 
  henttemperatur(); 
  skrivtemperatur(); 
  delay(1000); 
} 
/*-------------------------------------------- 
 *  Slut på programkontrol 
 ----------------------------------------------*/ 
 
 
 
 



/*-------------------------------------------- 
 *  Start på areal til Functions 
 ----------------------------------------------*/ 
 
void henttemperatur() 
{ 
  /* ADCH og ADCL i forlængelse af hinanden;*/ 
  double beggeregistre; 
 
/* Del 16 MHz CPU klok med 128 så vi får en AD klokke på 125 kHz */ 
ADCSRA |= (1 << ADPS2 ) | (1 << ADPS1 )| (1 << ADPS0); 
 
  /* Stil den interne  (1<< ADPS1) refernce til 1,1 volt og multiplexer på temperaturføleren  */ 
  ADMUX = (_BV(REFS1) | _BV(REFS0) | _BV(MUX3)) ; 
   
   
  /* Giv tid Giv tid så spændingen falder til ro */ 
  delay(20); 
 
  /* så skal vi have ADC enablet*/ 
  ADCSRA |= _BV(ADEN); 
   
  /* Giv tid Giv tid så ADC falder til ro  
  og data er stabile.*/ 
   delay(20);  
 
  ADCSRA |= _BV(ADSC);  /* start konvertering */ 
   
  /* vent på at konvertering er slut */ 
  while (bit_is_set(ADCSRA,ADSC));  
  delay(20); 
 
  /* Så læses registrene ADCH og ADCL 
   samt og at de kombineres til et ord */ 
  beggeregistre = (ADCL |(ADCH << 8)); 
 
  /* Så skal vi lige have korrigeret for startpunkt og for stejlhed */ 
  temperatur = ((beggeregistre - 332.2 ) / 1.06154 ); 
} 
 
void skrivtemperatur() 
{ 
lcd.setCursor(2,1);  /* start i 2. position, 2. linie */ 
lcd.print (temperatur,1); 
} 
 
/*-------------------------------------------- 
 *  Slut på areal til Functions og på programmet 
 ----------------------------------------------*/ 
 


